eman ta zabal zazu

Universidad del País Vasco | Euskal Herriko Unibertsitatea

Ejercicios resueltos

OCW 2020: *Parametrización y representación gráfica de superficies construidas*

Tema 2. Parametrización de curvas en el espacio

Equipo docente del curso
Martín Yagüe, Luis
Barrallo Calonge, Javier
Soto Merino, Juan Carlos
Lecubarri Alonso, Inmaculada

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I (EIB/BIE)
ETS de Arquitectura de Donostia-San Sebastián (ETSASS/DAGET)

EJERCICIOS DEL TEMA 2. PARAMETRIZACIÓN DE CURVAS EN EL ESPACIO

Ejercicio nº1

Enunciado

Sea la curva cerrada $C = C_1 \bigcup C_2 \bigcup C_3$ donde C_1 es un arco de circunferencia y C_2, C_3 son segmentos de recta tal como se muestra en la siguiente imagen:



Imagen 1. Curva C (imagen propia)

- a) Parametrice cada uno de los tramos de la curva C
- b) Represente gráficamente la curva C

Resolución

```
Remove["Global`*"] (* Borrado de variables utilizadas *)
```

- a) Parametrice cada uno de los tramos de la curva C
  - C_1 es el arco comprendido en el intervalo $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$ de una circunferencia de radio $R = 1$ y centro en el punto $P = (0, 1, 1)$

```
p = {0, 1, 1}; R = 1;
```

```
c1 = {x1 = p[[1]] + R*Cos[t], y1 = p[[2]] + R*Sin[t], z1 = 1}
```

```
{Cos[t], 1 + Sin[t], 1}
```

```
Solve[{x1 == 0, y1 == 2, z1 == 1}, t]
```

$$\left\{\left\{t \to \text{ConditionalExpression}\left[\frac{\pi}{2} + 2\pi C[1],\ C[1] \in \mathbb{Z}\right]\right\}\right\}$$

```
Solve[{x1 == 0, y1 == 0, z1 == 1}, t]
```

$$\left\{\left\{t \to \text{ConditionalExpression}\left[-\frac{\pi}{2} + 2\pi C[1],\ C[1] \in \mathbb{Z}\right]\right\}\right\}$$

$$C_1 \equiv \begin{cases} x(t) = & \cos(t) \\ y(t) = & 1 + \operatorname{sen}(t) \\ z(t) = & 1 \end{cases} \quad t \in \left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$$

- C_2 es el segmento del eje OZ comprendido en el intervalo $z \in [0, 1]$

```
c2 = {x2 = 0, y2 = 0, z2 = t}
```

```
{0, 0, t}
```

$$C_2 \equiv \begin{cases} x(t) = & 0 \\ y(t) = & 0 \\ z(t) = & t \end{cases} \quad t \in [0, 1]$$

- C_3 es el segmento de la recta que pasa por los puntos $O = (0, 0, 0)$ y $P_2 = (0, 2, 1)$ comprendido entre ambos puntos

```
{o = {0, 0, 0}, p2 = {0, 2, 1}}; v3 = p2 - o;
```

```
c3 = {x3 = o[[1]] + v3[[1]] t, y3 = o[[2]] + v3[[2]] t, z3 = o[[3]] + v3[[3]] t}
```

```
{0, 2 t, t}
```

$$C_3 \equiv \begin{cases} x(t) = & 0 \\ y(t) = & 2\,t \\ z(t) = & t \end{cases} \quad t \in [0, 1]$$

- b) Represente gráficamente la curva C
  - grafos de C_1, C_2, C_3

```
g1 = ParametricPlot3D[{x1, y1, z1}, {t, -Pi / 2, Pi / 2}, PlotStyle → {Blue, Thick}];
```

```
g2 = ParametricPlot3D[{x2, y2, z2}, {t, 0, 1}, PlotStyle → {Blue, Thick}];
```

```
g3 = ParametricPlot3D[{x3, y3, z3}, {t, 0, 1}, PlotStyle → {Blue, Thick}];
```

  - vértices de la línea C

```
vert = {o, p2, p1 = {0, 0, 1}};
```

```
cp = ListPointPlot3D[vert, PlotStyle → {Red, PointSize[Large]}];
```

  - unión de las gráficas anteriores

```
Show[g1, g2, g3, cp, PlotRange → All, AxesLabel → {X, Y, Z}]
```



Ejercicio nº2

Enunciado

Se considera el cono recto: $S \equiv x^2 + y^2 = z^2$

Parametrice y represente gráficamente cada una de las siguientes curvas planas:

- a) $S \bigcap \pi_1$ donde $\pi_1 \equiv z = 3$
- b) $S \bigcap \pi_2$ donde $\pi_2 \equiv z = 1 + x$
- c) $S \bigcap \pi_3$ donde $\pi_3 \equiv z = \frac{1}{2}(4 + x)$

Resolución

```
Remove["Global`*"]
```

- Se trata de curvas cónicas ya que se obtienen como la intersección del cono S y un plano

- Con la función **ContourPlot3D** se representan las superficies dadas para acompañar, posteriormente, las resoluciones analíticas con gráficas
  - cono recto: $S \equiv x^2 + y^2 = z^2$

```
cono = ContourPlot3D[x^2 + y^2 == z^2, {x, -5, 5},
   {y, -5, 5}, {z, -5, 5}, Mesh → None, ContourStyle → Opacity[0.5]];
```

  - plano: $\pi_1 \equiv z = 3$

```
pa = ContourPlot3D[z == 3, {x, -5, 5}, {y, -5, 5}, {z, -5, 5}, Mesh → None, ContourStyle → Opacity[0.25]];
```

  - plano: $\pi_2 \equiv z = 1 + x$

```
pb = ContourPlot3D[z == 1 + x, {x, -5, 5}, {y, -5, 5}, {z, -5, 5}, Mesh → None, ContourStyle → Opacity[0.25]];
```

  - plano: $\pi_3 \equiv z = \frac{1}{2}(4 + x)$

```
pc = ContourPlot3D[z == 2 + x/2, {x, -5, 5},
   {y, -5, 5}, {z, -5, 5}, Mesh → None, ContourStyle → Opacity[0.25]];
```

- a) $S \bigcap \pi_1 \equiv z = 3$
  - proyección ortogonal en el plano $z = 0$
    - eliminación de la variable z

```
Eliminate[{x^2 + y^2 == z^2, z == 3}, z] // Expand
```

```
9 - y^2 == x^2
```

    - cilindro proyectante: $x^2 + y^2 = 9$
    - curva proyección: $(C_a)_{z=0} \equiv \begin{cases} x^2 + y^2 = & 9 \\ z = & 0 \end{cases}$

- parametrización
  - de la curva de proyección: $(C_a)_{z=0} \equiv \begin{cases} x(t) = & 3\cos(t) \\ y(t) = & 3\,\text{sen}(t) \\ z(t) = & 0 \end{cases} \quad t \in [0, 2\pi]$

```
{xa[t_] = 3 Cos[t], ya[t_] = 3 Sin[t]}
```

```
{3 Cos[t], 3 Sin[t]}
```

  - de la cónica, C_a, donde $z = 3 \;\; \forall\, P \in C_a$:

```
za[t_] = 3;
```

$$\boxed{C_a \equiv \begin{cases} x(t) = & 3\cos(t) \\ y(t) = & 3\,\text{sen}(t) \\ z(t) = & 3 \end{cases} \quad t \in [0, 2\pi]}$$

  - la cónica es una circunferencia
- representación gráfica

```
Ca = ParametricPlot3D[{xa[t], ya[t], za[t]}, {t, 0, 2 Pi}, PlotStyle → {Green, Thickness[0.015]}];
```

```
Show[cono, pa, pa, Ca, AspectRatio → Automatic]
```



- b) $S \cap \pi_2 \equiv z = 1 + x$
  - proyección ortogonal en el plano $z = 0$
    - eliminación de la variable z

```
Eliminate[{x^2 + y^2 == z^2, z == 1 + x}, z] // Expand
```

```
y^2 == 1 + 2 x
```

  - cilindro proyectante: $y^2 - 2x = 1$
  - curva proyección: $(C_b)_{z=0} \equiv \begin{cases} y^2 - 2x = & 1 \\ z = & 0 \end{cases}$
- parametrización
  - curva proyección: $(C_b)_{z=0} \equiv \begin{cases} x(t) = & \frac{1}{2}(t^2 - 1) \\ y(t) = & t \\ z(t) = & 0 \end{cases} \quad t \in \mathbb{R}$

```
{xb[t_] = 1/2 (t^2 - 1), yb[t_] = t}
```

$$\left\{\frac{1}{2}\left(-1 + t^2\right), t\right\}$$

- de la cónica, C_b, tras obtener la coordenada z de la curva:

```
zb[t_] = 1 + xb[t] // Simplify
```

$\frac{1}{2}\left(1+t^2\right)$

$$C_b \equiv \begin{cases} x(t) = & \left(t^2-1\right)/2 \\ y(t) = & t \\ z(t) = & \left(t^2+1\right)/2 \end{cases} \quad t \in \mathbb{R}$$

- la cónica es una parábola
- representación gráfica

```
Cb = ParametricPlot3D[{xb[t], yb[t], zb[t]}, {t, -4, 4}, PlotStyle→{Green, Thickness[0.015]}];

Show[cono, pb, Cb, AspectRatio→Automatic]
```



- c) $S \cap \pi_3 \equiv z = \frac{1}{2}(4+x)$
  - proyección ortogonal en el plano $z = 0$
    - eliminación de la variable z

```
Eliminate[{x^2 + y^2 == z^2, z == 1/2 (4 + x)}, z] // Expand
```

$4y^2 == 16 + 8x - 3x^2$

```
m = CoefficientList[3*x^2 + 4*y^2 - 8*x - 16, {x, y}];

centro = {-m[[2, 1]]/2 / m[[3, 1]], -m[[1, 2]]/2 / m[[1, 3]]}
```

$\left\{\frac{4}{3}, 0\right\}$

```
k = (m[[2, 1]]/2)^2 / m[[3, 1]] + (m[[1, 2]]/2)^2 / m[[1, 3]] - m[[1, 1]];

If[m[[3, 1]] == m[[1, 3]], radio = Sqrt[k / m[[3, 1]]], semiejes = {Sqrt[k / m[[3, 1]]], Sqrt[k / m[[1, 3]]]}]
```

$\left\{\frac{8}{3}, \frac{4}{\sqrt{3}}\right\}$

- cilindro proyectante: $\frac{\left(x-\frac{4}{3}\right)^2}{\left(\frac{8}{3}\right)^2}+\frac{y^2}{\left(\frac{4}{\sqrt{3}}\right)^2}=1$
- curva proyección: $(C_c)_{z=0}\equiv\begin{cases}\frac{\left(x-\frac{4}{3}\right)^2}{\left(\frac{8}{3}\right)^2}+\frac{y^2}{\left(\frac{4}{\sqrt{3}}\right)^2}= & 1\\ z= & 0\end{cases}$

- parametrización

  - curva proyección: $(C_c)_{z=0}\equiv\begin{cases}x(t)= & \frac{4}{3}+\frac{8}{3}\cos(t)\\ y(t)= & 4\,\mathrm{sen}(t)\big/\sqrt{3} \quad t\in[0,2\pi]\\ z(t)= & 0\end{cases}$

```
{xc[t_] = 4/3 + 8/3 Cos[t], yc[t_] = 4/Sqrt[3] Sin[t]}
```

$\left\{\frac{4}{3}+\frac{8\,\mathrm{Cos}[t]}{3}, \frac{4\,\mathrm{Sin}[t]}{\sqrt{3}}\right\}$

- de la cónica, C_c, tras obtener la coordenada z de la curva:

```
zc[t_] = 1/2 (4 + xc[t]) // Simplify
```

$\frac{4}{3}(2+\mathrm{Cos}[t])$

$$\boxed{C_c\equiv\begin{cases}x(t)= & \frac{4}{3}+\frac{8}{3}\cos(t)\\ y(t)= & 4\,\mathrm{sen}(t)\big/\sqrt{3} \quad t\in\mathbb{R}\\ z(t)= & 4(2+\cos(t))/3\end{cases}}$$

- la cónica es una elipse

- representación gráfica

```
Cc = ParametricPlot3D[{xc[t], yc[t], zc[t]}, {t, -4, 4}, PlotStyle→{Green, Thickness[0.015]}];

Show[cono, pc, Cc, AspectRatio→Automatic]
```



Ejercicio nº3

Enunciado

Parametrice y represente gráficamente la curva: $C \equiv \begin{cases} x^2+y^2+z^2 = & 1 \\ \left(x-\frac{1}{2}\right)^2+y^2 = & \frac{1}{4} \end{cases}$



Imagen 2. Curva C (imagen propia)

Resolución

```
Remove["Global`*"]
```

- Haciendo uso de **ContourPlot3D** se representan las superficies cuya intersección es la curva C
  - esfera centrada en el origen y radio uno

```
esf = ContourPlot3D[x^2 + y^2 + z^2 == 1, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}, Mesh → False, Boxed → False,
    AxesLabel → {X, Y, Z}, ContourStyle → {LightBlue, Opacity[0.7]}, AspectRatio → Automatic];
```

  - cilindro circular recto de radio $R = \frac{1}{2}$ y eje en $e \equiv \begin{cases} x = 1 \\ y = 0 \end{cases}$

```
cil = ContourPlot3D[(x - 1/2)^2 + y^2 == 1/4, {x, 0, 1}, {y, -1/2, 1/2}, {z, -1, 1}, Mesh → False,
    Boxed → False, AxesLabel → {X, Y, Z}, ContourStyle → Opacity[0.5], AspectRatio → Automatic];
```

  - unión de las gráficas

```
g1 = Show[esf, cil, PlotRange → All, AspectRatio → Automatic]
```



- Proyección ortogonal en el plano $z = 0$
  - eliminación de la variable z

```
Eliminate[{x^2 + y^2 + z^2 == 1, (x - 1/2)^2 + y^2 == 1/4}, z] // Expand
```

```
y^2 == x - x^2
```

- cilindro proyectante: $x^2 - x + y^2 = 0 \Rightarrow \left(x - \frac{1}{2}\right)^2 + y^2 = \frac{1}{4}$
- curva proyección: $(C)_{z=0} \equiv \begin{cases} \left(x - \frac{1}{2}\right)^2 + y^2 = & \frac{1}{4} \\ z = & 0 \end{cases}$

- Parametrización
  - curva proyección: $(C)_{z=0} \equiv \begin{cases} x(t) = & \frac{1}{2} + \frac{1}{2}\cos(t) \\ y(t) = & \operatorname{sen}(t)/2 \\ z(t) = & 0 \end{cases} \quad t \in [0, 2\pi]$

```
{x[t_] = 1/2 + 1/2 Cos[t], y[t_] = 1/2 Sin[t]}
```

$\left\{\frac{1}{2} + \frac{\mathrm{Cos}[t]}{2}, \frac{\mathrm{Sin}[t]}{2}\right\}$

- obtención de la coordenada z de la curva C:

```
Solve[x^2 + y^2 + z^2 == 1, z]
```

$\left\{\left\{z \to -\sqrt{1 - x^2 - y^2}\right\}, \left\{z \to \sqrt{1 - x^2 - y^2}\right\}\right\}$

- rama positiva: $\boxed{C_p \equiv \begin{cases} x(t) = & \frac{1}{2} + \frac{1}{2}\cos(t) \\ y(t) = & \operatorname{sen}(t)/2 \\ z(t) = & \operatorname{sen}(t/2) \end{cases} \quad t \in [0, 2\pi]}$

```
zp[t_] = Sqrt[1 - x^2 - y^2] /. {x -> 1/2 + 1/2 Cos[t], y -> 1/2 Sin[t]} // Simplify
```

$\sqrt{\mathrm{Sin}\left[\frac{t}{2}\right]^2}$

- rama negativa: $\boxed{C_n \equiv \begin{cases} x(t) = & \frac{1}{2} + \frac{1}{2}\cos(t) \\ y(t) = & \operatorname{sen}(t)/2 \\ z(t) = & -\operatorname{sen}(t/2) \end{cases} \quad t \in [0, 2\pi]}$

```
zn[t_] = -Sqrt[1 - x^2 - y^2] /. {x -> 1/2 + 1/2 Cos[t], y -> 1/2 Sin[t]} // Simplify
```

$-\sqrt{\mathrm{Sin}\left[\frac{t}{2}\right]^2}$

- Representación gráfica
  - plano de proyección $z = 0$

```
z0 = ContourPlot3D[z == 0, {x, 0, 1}, {y, -1, 1}, {z, -2, 2}, Mesh → False, Boxed → False,
   AxesLabel → {X, Y, Z}, ContourStyle → Directive[Pink, Opacity[0.2], Specularity[White, 30]]];
```

  - proyección ortogonal $(C)_{z=0}$

```
proyz0 = ParametricPlot3D[{x[t], y[t], 0}, {t, 0, 2 Pi}, PlotStyle → {Red, Thickness[0.015]}];
```

  - curva C

```
Cpos = ParametricPlot3D[{x[t], y[t], zp[t]}, {t, 0, 2 Pi}, PlotStyle → {Green, Thickness[0.015]}];
```

```
Cneg = ParametricPlot3D[{x[t], y[t], zn[t]}, {t, 0, 2 Pi}, PlotStyle→{Green, Thickness[0.015]}];

Show[Cpos, Cneg]
```



- curva C, junto con su proyección ortogonal y el cilindro proyectante

```
Show[cil, z0, Cpos, Cneg, proyz0, AspectRatio→Automatic]
```



- curva C, junto con la esfera y el cilindro

```
Show[cil, esf, Cpos, Cneg, PlotRange→All, AspectRatio→Automatic]
```



Ejercicio nº4

Enunciado

La bóveda de cañón es un elemento constructivo del románico. Se erige concatenando de forma longitudinal arcos de medio punto (semicircunferencias). Geométricamente, se trata de medio cilindro circular.



Imagen 3. Generación del arco de medio punto (imagen propia)

La bóveda de arista es el elemento arquitectónico del románico que se usa para techar espacios cuadrangulares. Resulta de intersectar ortogonalmente dos bóvedas de cañón del mismo radio.



Imagen 4. Generación de la bóveda de arista (imagen propia)

Parametrice y represente las líneas que definen los seis arcos de medio punto y las dos aristas de la bóveda si se quiere cubrir un espacio cuadrangular de 10 metros de lado.

Resolución

```
Remove["Global`*"]
```

- La bóveda resulta de la intersección de dos superficies cilíndricas.
- Se establece un sistema cartesiano de referencia tal que el plano OXY contiene los puntos de arranque de los seis arcos, los ejes OX y OY coinciden con las proyecciones sobre OXY de las aristas y el eje OZ es la recta vertical que pasa por la clave (dovela central).
- El radio de las circunferencias que generan los arcos de medio punto es: $R = 5\,\mathrm{m}$

```
r = 5;
```

- Se establece una sola parametrización que define la base, $x(t)$, y la altura de uno de los arcos, $z(t)$; el resto de curvas se obtiene a partir de ella

```
{x[t_] = r Cos[t], z[t_] = r Sin[t]}
```

```
{5 Cos[t], 5 Sin[t]}
```

- Se representa en el plano uno de los arcos

```
ParametricPlot[{{x[t], z[t]}}, {t, 0, Pi}, PlotStyle→{Brown, Thick}]
```



- Ahora hay que representar en el espacio ese tipo de arcos:
  - arcos en los muros, son paralelos y distan 10 metros (se trazan en los planos $y = -5\,\mathrm{m}$, $y = 5\,\mathrm{m}$)

```
arc1 =
 ParametricPlot3D[{{x[t], -5, z[t]}, {x[t], 5, z[t]}}, {t, 0, Pi}, PlotStyle → {Brown, Brown, Thick}]
```



- arcos en la nave, son paralelos y distan 10 metros (se trazan en los planos $x = -5\,\mathrm{m}$, $x = 5\,\mathrm{m}$)

```
arc2 =
 ParametricPlot3D[{{-5, x[t], z[t]}, {5, x[t], z[t]}}, {t, 0, Pi}, PlotStyle → {Brown, Brown, Thick}]
```



- representación conjunta de estos cuatro arcos

```
Show[arc1, arc2, PlotRange → All, Boxed → True, Axes → True]
```



- arcos cruzados (ubicados en los planos $x = -y$, $x = y$)

```
arc3 = ParametricPlot3D[{x[t], x[t], z[t]}, {t, 0, Pi}, PlotStyle → {Brown, Brown, Thick}];

arc4 = ParametricPlot3D[{x[t], -x[t], z[t]}, {t, 0, Pi}, PlotStyle → {Brown, Brown, Thick}];
```

- Representación de las aristas

```
aristas =
  ParametricPlot3D[{{t, 0, z[Pi/2]}, {0, t, z[Pi/2]}}, {t, -5, 5}, PlotStyle → {Brown, Brown, Thick}];
```

- Representación gráfica de todo el conjunto de arcos y aristas

```
Show[arc1, arc2, arc3, arc4, aristas, PlotRange → All, Boxed → False, Axes → False]
```



Ejercicio nº5

Enunciado

Parametrice y represente gráficamente cada una de las siguientes hélices $\forall\, t \in [0, 4\pi]$:

- a) $H_1 \equiv \begin{cases} x(t) = & \sqrt{2}\cos(t) \\ y(t) = & \sqrt{2}\,\mathrm{sen}(t) \\ z(t) = & t \end{cases}$

- b) $H_2 \equiv \begin{cases} x(t) = & \sqrt{2}\cos(2t) \\ y(t) = & \sqrt{2}\,\mathrm{sen}(2t) \\ z(t) = & t \end{cases}$

- c) $H_3 \equiv \begin{cases} x(t) = & \sqrt{2}\cos(t) \\ y(t) = & -\sqrt{2}\,\mathrm{sen}(t) \\ z(t) = & t \end{cases}$

Resolución

```
Remove["Global`*"]
```

- Con **ContourPlot3D** se representa la superficie cilíndrica, las hélices se trazarán sobre el cilindro
- Se trata de un cilindro recto de radio: $R = \sqrt{2}$

```
cil = ContourPlot3D[x^2 + y^2 == 2, {x, -√2, √2}, {y, -√2, √2}, {z, 0, 4 Pi}, Mesh → False,
   Boxed → False, Axes → False, ContourStyle → {LightGray, Opacity[0.7]}, AspectRatio → Automatic];
```

- a) $H_1 \equiv \begin{cases} x(t) = & \sqrt{2}\cos(t) \\ y(t) = & \sqrt{2}\,\mathrm{sen}(t) \\ z(t) = & t \end{cases}$

  - parametrización

```
h1 = {√2 Cos[t], √2 Sin[t], t};
```

  - representación gráfica
    - cuando $t \in [0, 2\pi]$

```
H11 = ParametricPlot3D[h1, {t, 0, 2 Pi}, PlotStyle → {Green, Thickness[0.015]}];
```

    - cuando $t \in [2\pi, 4\pi]$

```
H12 = ParametricPlot3D[h1, {t, 2 Pi, 4 Pi}, PlotStyle → {Blue, Thickness[0.015]}];
```

```
Show[cil, H11, H12, PlotRange → All]
```



- la hélice H_1 da una vuelta alrededor del cilindro cada 2π radianes
- paso de la hélice H_1: $p_1 = 2\pi$

```
inic = h1 /. t → 0
```

$\{\sqrt{2}, 0, 0\}$

```
Solve[{h1[[1]] == inic[[1]], h1[[2]] == inic[[2]]}, t]
```

$\{\{t \to \text{ConditionalExpression}[2\pi C[1], C[1] \in \mathbb{Z}]\}\}$

- b) $H_2 \equiv \begin{cases} x(t) = & \sqrt{2}\cos(2t) \\ y(t) = & \sqrt{2}\,\text{sen}(2t) \\ z(t) = & t \end{cases}$
  - parametrización

```
h2 = {√2 Cos[2 t], √2 Sin[2 t], t}
```

$\{\sqrt{2}\,\text{Cos}[2t], \sqrt{2}\,\text{Sin}[2t], t\}$

- representación gráfica
  - cuando $t \in [0, 2\pi]$

```
H21 = ParametricPlot3D[h2, {t, 0, 2 Pi}, PlotStyle → {Green, Thickness[0.015]}];
```

  - cuando $t \in [2\pi, 4\pi]$

```
H22 = ParametricPlot3D[h2, {t, 2 Pi, 4 Pi}, PlotStyle → {Blue, Thickness[0.015]}];
Show[cil, H21, H22, PlotRange → All]
```



- la hélice H_2 da dos vuelta alrededor del cilindro cada 2π radianes
- paso de la hélice H_2: $p_2 = \pi$

```
inic = h2 /. t → 0
```

$\{\sqrt{2}, 0, 0\}$

```
Solve[{h2[[1]] == inic[[1]], h2[[2]] == inic[[2]]}, t]
```

$\{\{t \to \text{ConditionalExpression}[\pi C[1], C[1] \in \mathbb{Z}]\}\}$

- c) $H_3 \equiv \begin{cases} x(t) = & \sqrt{2}\cos(t) \\ y(t) = & -\sqrt{2}\,\text{sen}(t) \\ z(t) = & t \end{cases}$
  - parametrización

```
h3 = {√2 Cos[t], -√2 Sin[t], t};
```

- representación gráfica
  - cuando $t \in [0, 2\pi]$

```
H31 = ParametricPlot3D[h3, {t, 0, 2 Pi}, PlotStyle → {Red, Thickness[0.015]}];
```

  - cuando $t \in [2\pi, 4\pi]$

```
H32 = ParametricPlot3D[h3, {t, 2 Pi, 4 Pi}, PlotStyle → {Purple, Thickness[0.015]}];
```

```
Show[cil, H31, H32, PlotRange → All]
```



- la hélice H_3 da una vuelta alrededor del cilindro cada 2π radianes aunque invierte el sentido del giro respecto de H_1
- paso de la hélice H_3: $p_3 = 2\pi$

Ejercicio nº6

Enunciado

Parametrice y represente gráficamente cada una de las siguientes hélices:

- a) $H_1 \equiv \begin{cases} x(t) = & 2(t-2\pi)\cdot\cos(t)/(2\pi) \\ y(t) = & 2(t-2\pi)\cdot\operatorname{sen}(t)/(2\pi) \\ z(t) = & 4(t/2\pi) \end{cases}$
- b) $H_2 \equiv \begin{cases} x(t) = & 2(t-2\pi)\cdot\cos(2t)/(2\pi) \\ y(t) = & 2(t-2\pi)\cdot\operatorname{sen}(2t)/(2\pi) \\ z(t) = & 4(t/2\pi) \end{cases}$
- c) $H_3 \equiv \begin{cases} x(t) = & 2(t-2\pi)\cdot\cos(3t)/(2\pi) \\ y(t) = & 2(t-2\pi)\cdot\operatorname{sen}(3t)/(2\pi) \\ z(t) = & 4(t/2\pi) \end{cases}$

Resolución

```
Remove["Global`*"]
```

- Con la función **ContourPlot3D** se representa la superficie cónica, las hélices se trazarán sobre el cono
- Se trata de la mitad negativa de un cono recto de radio en la base $R = 2$ y altura $h = 4$; el vértice se sitúa en $V = (0, 0, 4)$ para que la base se encuentre en el plano $z = 0$ (cota 0)

```
con = ContourPlot3D[x^2 + y^2 == (z - 4)^2/4, {x, -4, 4}, {y, -4, 4}, {z, 0, 4}, Mesh → False, Boxed → False,
   Axes → False, ContourStyle → {LightGray, Opacity[0.7]}, AspectRatio → Automatic];
```

- a) $H_1 \equiv \begin{cases} x(t) = & 2\,(t - 2\,\pi)\cdot \cos(t)/(2\,\pi) \\ y(t) = & 2\,(t - 2\,\pi)\cdot \mathrm{sen}(t)/(2\,\pi) \\ z(t) = & 4\,(t/2\,\pi) \end{cases}$
  - parametrización

```
h1 = {2 (t - 2 Pi) Cos[t] / (2 Pi), 2 (t - 2 Pi) Sin[t] / (2 Pi), 4 t / (2 Pi)};
```

  - representación gráfica
    - cuando $t \in [0, 2\,\pi]$

```
H11 = ParametricPlot3D[h1, {t, 0, 2 Pi}, PlotStyle → {Green, Thickness[0.015]}];
```

```
Show[con, H11, PlotRange → All]
```



  - la hélice H_1 da una vuelta alrededor del cilindro cada 2π radianes
- b) $H_2 \equiv \begin{cases} x(t) = & 2\,(t - 2\,\pi)\cdot \cos(2\,t)/(2\,\pi) \\ y(t) = & 2\,(t - 2\,\pi)\cdot \mathrm{sen}(2\,t)/(2\,\pi) \\ z(t) = & 4\,(t/2\,\pi) \end{cases}$
  - parametrización

```
h2 = {2 (t - 2 Pi) Cos[2 t] / (2 Pi), 2 (t - 2 Pi) Sin[2 t] / (2 Pi), 4 t / (2 Pi)};
```

  - representación gráfica
    - cuando $t \in [0, \pi]$

```
H21 = ParametricPlot3D[h2, {t, 0, Pi}, PlotStyle → {Green, Thickness[0.015]}];
```

    - cuando $t \in [\pi, 2\,\pi]$

```
H22 = ParametricPlot3D[h2, {t, Pi, 2 Pi}, PlotStyle → {Blue, Thickness[0.015]}];
```

```
Show[con, H21, H22, PlotRange → All]
```



- c) $H_3 \equiv \begin{cases} x(t) = & 2(t-2\pi)\cdot\cos(3t)/(2\pi) \\ y(t) = & 2(t-2\pi)\cdot\text{sen}(3t)/(2\pi) \\ z(t) = & 4(t/2\pi) \end{cases}$
  - parametrización

```
h3 = {2 (t - 2 Pi) Cos[3 t] / (2 Pi), 2 (t - 2 Pi) Sin[3 t] / (2 Pi), 4 t / (2 Pi)};
```

  - representación gráfica
    - cuando $t \in \left[0, \frac{2\pi}{3}\right]$

```
H31 = ParametricPlot3D[h3, {t, 0, 2 Pi / 3}, PlotStyle → {Green, Thickness[0.015]}];
```

    - cuando $t \in \left[\frac{2\pi}{3}, \frac{4\pi}{3}\right]$

```
H32 = ParametricPlot3D[h3, {t, 2 Pi / 3, 4 Pi / 3}, PlotStyle → {Blue, Thickness[0.015]}];
```

    - cuando $t \in \left[\frac{4\pi}{3}, 2\pi\right]$

```
H33 = ParametricPlot3D[h3, {t, 4 Pi / 3, 2 Pi}, PlotStyle → {Red, Thickness[0.015]}];

Show[con, H31, H32, H33, PlotRange → All]
```